#被动技能【飓风旋律】
execute as @a[tag=BOSSxiu,x_rotation=-30..-10] run effect give @s levitation 1 0 true
execute as @a[tag=BOSSxiu,x_rotation=-45..-31] run effect give @s levitation 1 1 true
execute as @a[tag=BOSSxiu,x_rotation=-60..-46] run effect give @s levitation 1 3 true
execute as @a[tag=BOSSxiu,x_rotation=-90..-61] run effect give @s levitation 1 5 true

execute as @a[tag=BOSSxiu,x_rotation=35..60,scores={skJump=1..}] if data entity @s {SelectedItem:{id:"minecraft:feather",Count:1b}} run effect give @s levitation 1 3
execute as @a[tag=BOSSxiu,x_rotation=61..90,scores={skJump=1..}] if data entity @s {SelectedItem:{id:"minecraft:feather",Count:1b}} run effect give @s levitation 1 6
execute as @a[tag=BOSSxiu,x_rotation=35..90,scores={skJump=1..}] at @s if data entity @s {SelectedItem:{id:"minecraft:feather",Count:1b}} run particle minecraft:spit ~ ~ ~ 0.7 0.2 0.7 0.05 10 normal @a
scoreboard players reset @a[tag=BOSSxiu,scores={skJump=1..}] skJump
# execute as @a[tag=BOSSxiu] at @s run effect clear @s levitation
#被动技能【神虑之风】
execute as @a[tag=BOSSxiu] at @s if block ~ ~-1 ~ air if block ~ ~-2 ~ air if block ~ ~-3 ~ air unless block ~ ~-5 ~ air run effect give @s slow_falling 1 0
execute as @a[tag=BOSSxiu] at @s if block ~ ~-1 ~ air if block ~ ~-2 ~ air if block ~ ~-3 ~ air unless block ~ ~-6 ~ air run effect give @s slow_falling 1 0
execute as @a[tag=BOSSxiu] at @s if block ~ ~-1 ~ air if block ~ ~-2 ~ air if block ~ ~-3 ~ air unless block ~ ~-7 ~ air run effect give @s slow_falling 1 0
#execute as @a[tag=BOSSxiu,x_rotation=90] if data entity @s {SelectedItem:{id:"minecraft:feather",Count:1b}} run effect clear @s levitation

#被动技能【狂风之力】
execute as @a[tag=BOSSxiu,nbt={OnGround:0b}] at @s if block ~ ~-1.5 ~ air run item replace entity @s armor.head with glass{AttributeModifiers: [{Operation: 0, Amount: +2.5d, UUID: [I; 678923278, 482171669, -276225087, -883835081], Slot: head, AttributeName: "generic.attack_damage", Name: "generic.attackDamage"}], display: {Name: '[{"text":"狂风之力","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]', Lore: ['{"text":"被动技能【狂风之力】","color":"dark_purple","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}', '{"text":"休在天空时获得+2.5攻击力+0.4攻击速度","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}']}, Enchantments: [{id: "minecraft:loyalty", lvl: 10}]} 1
clear @a[tag=BOSSxiu,nbt={OnGround:1b}] glass

execute as @a[tag=BOSSxiu,scores={skShangHai=1..}] if data entity @s {SelectedItem:{id:"minecraft:feather",Count:1b}} at @s rotated ~ 0 positioned ^ ^ ^3 as @p[tag=!BOSSxiu,distance=..3] at @s if block ~ ~-0.5 ~ air run effect give @s levitation 3 1
execute as @a[tag=BOSSxiu,scores={skShangHai=1..}] if data entity @s {SelectedItem:{id:"minecraft:feather",Count:1b}} at @s rotated ~ 0 positioned ^ ^ ^3 as @p[tag=!BOSSxiu,distance=..3] at @s unless block ~ ~-0.5 ~ air run tp @s ~ ~1.5 ~
execute as @a[tag=BOSSxiu,scores={skShangHai=1..}] if data entity @s {SelectedItem:{id:"minecraft:feather",Count:1b}} at @s rotated ~ 0 positioned ^ ^ ^3 as @p[tag=!BOSSxiu,distance=..3] at @s unless block ~ ~-0.5 ~ air run effect give @s levitation 1 0
scoreboard players reset @a[tag=BOSSxiu,scores={skShangHai=1..}] skShangHai
#被动技能【旋风护体】
execute as @a[tag=BOSSxiu] run effect give @s resistance 1 0 true 
execute as @a[tag=BOSSxiu] at @s if block ~ ~-1.5 ~ air run effect give @s minecraft:resistance 1 1 true






#武器技能【狂风席卷】改变箭
execute at @a[tag=BOSSxiu,scores={skBow1=1..}] run tag @e[type=spectral_arrow,distance=0..3] add kuangfeng
execute at @a[tag=BOSSxiu,scores={skBow0=1..}] as @e[distance=0..4,type=spectral_arrow,tag=kuangfeng] run data merge entity @s {NoGravity:1b}
#武器技能【狂风席卷】
execute as @a[tag=BOSSxiu,scores={skBow1=1..}] at @s run function nw:yingxiong/0boss/02xiu/skill/0kuangfeng
execute at @a[tag=BOSSxiu,scores={skBow1=1..}] run tag @e[type=spectral_arrow,distance=0..3] add kuangfeng
#武器技能【狂风席卷】CD
execute as @a[tag=BOSSxiu,scores={skCDPearl=0}] at @s run function nw:yingxiong/0boss/02xiu/skill/0kuangfeng_cd

#武器技能【狂风席卷】跟随效果
execute as @e[type=spectral_arrow,tag=kuangfeng] at @s run particle minecraft:dust 0.2 0.2 0.2 1 ~ ~ ~ 1.2 1.5 1.2 0.1 20 normal @a
execute at @e[type=spectral_arrow,tag=kuangfeng] run effect give @e[distance=0..4,tag=!BOSSxiu] levitation 2 3
execute at @e[type=spectral_arrow,tag=kuangfeng] run kill @e[tag=yasuo,type=armor_stand,distance=0..4]
execute at @e[type=spectral_arrow,tag=kuangfeng] run kill @e[tag=!kuangfeng,type=#arrows,distance=..4]
execute at @e[type=spectral_arrow,tag=kuangfeng] run kill @e[type=fireball,distance=0..4]
execute at @e[type=spectral_arrow,tag=kuangfeng] run kill @e[type=snowball,distance=0..4]
#[狂风]落地时生成[小型狂风龙卷]
execute at @e[type=spectral_arrow,tag=kuangfeng,nbt={inGround:1b}] run effect give @a[tag=!BOSSxiu,distance=..5] levitation 1 5 true
execute at @e[type=spectral_arrow,tag=kuangfeng,nbt={inGround:1b}] run effect give @e[tag=!BOSSxiu,distance=..5] instant_damage 1 0
execute at @e[type=spectral_arrow,tag=kuangfeng,nbt={inGround:1b}] run summon minecraft:armor_stand ~ ~ ~ {Tags:[kuangfeng],Invulnerable:1b,Marker:1b,PersistenceRequired:1b,NoGravity:1b,OnGround:1b,Invisible:1,Small:1,NoBasePlate:1,Rotation:[0f],Pose:{Head:[0f,0f,0f],Body:[0f,0f,0f],LeftArm:[0f,0f,0f],RightArm:[0f,0f,0f],LeftLeg:[0f,0f,0f],RightLeg:[0f,0f,0f]}}
kill @e[type=spectral_arrow,tag=kuangfeng,nbt={inGround:1b}]


#[小型狂风龙卷]休自身效果
execute at @e[type=minecraft:armor_stand,tag=kuangfeng] run effect give @a[tag=BOSSxiu,distance=..5] speed 1 0 true
#[小型狂风龙卷]其他玩家减速效果
execute at @e[type=minecraft:armor_stand,tag=kuangfeng] run effect give @e[tag=!BOSSxiu,distance=..5] slowness 2 1 true

#[小型狂风龙卷]中心特效
execute as @e[type=armor_stand,tag=kuangfeng] at @s run particle dust 0.2 0.2 0.2 1 ~ ~0.5 ~ 3 0.1 3 0.01 15 normal @a
execute at @e[type=armor_stand,tag=kuangfeng] positioned ~ ~0.6 ~ run particle large_smoke ~ ~ ~ 0.4 0.5 0.4 0.05 1 normal @a
execute at @e[type=armor_stand,tag=kuangfeng] positioned ~ ~1.7 ~ run particle large_smoke ~ ~ ~ 0.8 0.5 0.8 0.05 2 normal @a
execute at @e[type=armor_stand,tag=kuangfeng] positioned ~ ~3.2 ~ run particle large_smoke ~ ~ ~ 1.2 0.5 1.2 0.05 3 normal @a
execute at @e[type=armor_stand,tag=kuangfeng] positioned ~ ~4.5 ~ run particle large_smoke ~ ~ ~ 1.5 0.5 1.5 0.05 5 normal @a
kill @e[type=armor_stand,tag=kuangfeng,scores={skTimeKe=100..}]


#主动技能【狂风休止】
execute as @a[tag=BOSSxiu,scores={skEgg=1..}] at @s run function nw:yingxiong/0boss/02xiu/skill/1jinglan
#主动技能【狂风休止】CD
execute as @a[tag=BOSSxiu,scores={skCDEgg=0}] at @s run function nw:yingxiong/0boss/02xiu/skill/1jinglan_cd




#主动技能【龙卷风暴】
execute as @a[tag=BOSSxiu,scores={skSnowball=1..}] at @s run function nw:yingxiong/0boss/02xiu/skill/2longjuanfeng
#主动技能【龙卷风暴】CD
execute as @a[tag=BOSSxiu,scores={skCDSnowball=0}] run function nw:yingxiong/0boss/02xiu/skill/2longjuanfeng_cd
#主动技能[狂放区域]两秒后形成[龙卷风暴]
execute at @e[type=minecraft:armor_stand,tag=kuangfengquyu,tag=!finish,scores={skTimeKe=40..}] unless entity @e[type=armor_stand,tag=kuangfeng,distance=..6] run summon minecraft:armor_stand ~ ~ ~ {Tags:[BOSSxiu],CustomName:'[{"text":"龙卷风暴","color":"green","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]',CustomNameVisible:1b,Invulnerable:1b,Marker:1b,PersistenceRequired:1b,NoGravity:1b,OnGround:1b,Invisible:1,Small:1,NoBasePlate:1,Rotation:[0f],Pose:{Head:[0f,0f,0f],Body:[0f,0f,0f],LeftArm:[0f,0f,0f],RightArm:[0f,0f,0f],LeftLeg:[0f,0f,0f],RightLeg:[0f,0f,0f]}}
execute at @e[type=minecraft:armor_stand,tag=kuangfengquyu,tag=!finish,scores={skTimeKe=40..}] if entity @e[type=armor_stand,tag=kuangfeng,distance=..6] run summon minecraft:armor_stand ~ ~ ~ {Tags:[BOSSxiu,kuangfeng],CustomName:'[{"text":"大型狂风龙卷","color":"green","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]',CustomNameVisible:1b,Invulnerable:1b,Marker:1b,PersistenceRequired:1b,NoGravity:1b,OnGround:1b,Invisible:1,Small:1,NoBasePlate:1,Rotation:[0f],Pose:{Head:[0f,0f,0f],Body:[0f,0f,0f],LeftArm:[0f,0f,0f],RightArm:[0f,0f,0f],LeftLeg:[0f,0f,0f],RightLeg:[0f,0f,0f]}}
execute at @e[type=minecraft:armor_stand,tag=kuangfengquyu,tag=!finish,scores={skTimeKe=40..}] run kill @e[type=armor_stand,tag=!BOSSxiu,tag=kuangfeng,distance=..6]
tag @e[type=minecraft:armor_stand,tag=kuangfengquyu,tag=!finish,scores={skTimeKe=40..}] add finish


#[龙卷风暴]被狂风扰乱形成大型[狂风龙卷]
execute as @e[type=armor_stand,tag=BOSSxiu] at @s positioned ~ ~2 ~ if entity @e[type=spectral_arrow,tag=kuangfeng,distance=0..3] run tag @s add a
execute as @e[type=armor_stand,tag=BOSSxiu] at @s positioned ~ ~2 ~ run kill @e[type=spectral_arrow,tag=kuangfeng,distance=0..3]
execute at @e[type=armor_stand,tag=BOSSxiu,tag=a] run effect give @a[tag=!BOSSxiu,distance=..6] levitation 1 6 true
execute at @e[type=armor_stand,tag=BOSSxiu,tag=a] run effect give @a[tag=!BOSSxiu,distance=..6] instant_damage 1 0 true
execute at @e[type=armor_stand,tag=BOSSxiu,tag=a] run effect give @a[tag=!BOSSxiu,distance=..6] wither 3 1 true
execute at @e[type=armor_stand,tag=BOSSxiu,tag=a] run particle large_smoke ~ ~0.5 ~ 0.4 0.5 0.4 0.05 10 normal @a
execute at @e[type=armor_stand,tag=BOSSxiu,tag=a] run particle large_smoke ~ ~1.5 ~ 0.8 0.5 0.8 0.05 20 normal @a
execute at @e[type=armor_stand,tag=BOSSxiu,tag=a] run particle large_smoke ~ ~3 ~ 1.2 0.5 1.2 0.05 30 normal @a
execute at @e[type=armor_stand,tag=BOSSxiu,tag=a] run particle large_smoke ~ ~4.5 ~ 1.6 0.5 1.6 0.05 40 normal @a
execute at @e[type=armor_stand,tag=BOSSxiu,tag=a] run particle large_smoke ~ ~6 ~ 1.6 0.5 1.6 0.05 50 normal @a
execute at @e[type=armor_stand,tag=BOSSxiu,tag=a] run particle large_smoke ~ ~7.5 ~ 2 0.5 2 0.05 60 normal @a
kill @e[type=armor_stand,tag=BOSSxiu,tag=a]



#[狂放区域]中自身移速效果
execute at @e[type=armor_stand,tag=kuangfengquyu] run effect give @a[tag=BOSSxiu,distance=0..3] speed 1 2 false
execute at @e[type=armor_stand,tag=kuangfengquyu] run effect give @a[tag=BOSSxiu,distance=3..6] speed 1 1 false
execute at @e[type=armor_stand,tag=kuangfengquyu] run effect give @a[tag=BOSSxiu,distance=0..9] speed 1 0 false
#[狂放区域]中其他玩家的缓慢与漂浮
execute at @e[type=minecraft:armor_stand,tag=kuangfengquyu] run effect give @e[distance=0..2,tag=!BOSSxiu] slowness 1 2 false
execute at @e[type=minecraft:armor_stand,tag=kuangfengquyu] run effect give @e[distance=2..4,tag=!BOSSxiu] slowness 1 1 false
execute at @e[type=minecraft:armor_stand,tag=kuangfengquyu] run effect give @e[distance=4..6,tag=!BOSSxiu] slowness 1 0 false
#[狂放区域]显示范围
execute at @e[type=armor_stand,tag=kuangfengquyu] run particle enchant ~ ~0.5 ~ 4 0.1 4 0.01 20 normal @a
execute as @e[type=armor_stand,tag=kuangfengquyu] at @s run tp @s ~ ~ ~ ~4 ~
execute at @e[type=armor_stand,tag=kuangfengquyu] rotated ~ 0 run particle firework ^ ^0.5 ^9 0.1 0.1 0.1 0.01 1 normal @a
execute at @e[type=armor_stand,tag=kuangfengquyu] rotated ~ 0 run particle firework ^ ^0.5 ^-9 0.1 0.1 0.1 0.01 1 normal @a
execute as @e[type=armor_stand,tag=kuangfengquyu] at @s run tp @s ~ ~ ~ ~4 ~
execute at @e[type=armor_stand,tag=kuangfengquyu] rotated ~ 0 run particle firework ^ ^0.5 ^9 0.1 0.1 0.1 0.01 1 normal @a
execute at @e[type=armor_stand,tag=kuangfengquyu] rotated ~ 0 run particle firework ^ ^0.5 ^-9 0.1 0.1 0.1 0.01 1 normal @a



#[龙卷风暴]缓慢向休移动
execute as @e[type=armor_stand,tag=BOSSxiu] at @s if entity @a[tag=BOSSxiu,distance=0..7] facing entity @a[tag=BOSSxiu,limit=1] feet run tp @s ^ ^ ^0.06
#[龙卷风暴]把其他实体向中心拉扯
execute as @e[tag=!kuangfeng,tag=!BOSSxiu,tag=!kuangfengquyu] at @s facing entity @e[type=minecraft:armor_stand,tag=BOSSxiu,scores={skTimeKe=15..},limit=1,distance=6.1..9] feet run tp @s ^ ^ ^0.3
execute as @e[tag=!kuangfeng,tag=!BOSSxiu,tag=!kuangfengquyu] at @s facing entity @e[type=minecraft:armor_stand,tag=BOSSxiu,scores={skTimeKe=15..},limit=1,distance=3.1..6] feet run tp @s ^ ^ ^0.4
execute as @e[tag=!kuangfeng,tag=!BOSSxiu,tag=!kuangfengquyu] at @s facing entity @e[type=minecraft:armor_stand,tag=BOSSxiu,scores={skTimeKe=15..},limit=1,distance=..3] feet run tp @s ^ ^ ^0.5
scoreboard players reset @e[type=minecraft:armor_stand,tag=BOSSxiu,scores={skTimeKe=17..}] skTimeKe

#[龙卷风暴]休自身效果
execute at @e[type=armor_stand,tag=BOSSxiu] run effect give @a[tag=BOSSxiu,distance=0..7] slow_falling 1 0
#[龙卷风暴]其他实体漂浮
execute at @e[type=minecraft:armor_stand,tag=BOSSxiu,tag=!kuangfeng] run effect give @e[distance=0..2,tag=!BOSSxiu] levitation 1 3 true
execute at @e[type=minecraft:armor_stand,tag=BOSSxiu,tag=!kuangfeng] run effect give @e[distance=2..4,tag=!BOSSxiu] levitation 1 2 true
execute at @e[type=minecraft:armor_stand,tag=BOSSxiu,tag=!kuangfeng] run effect give @e[distance=4..6,tag=!BOSSxiu] levitation 1 1 true

#[龙卷风暴]特效
execute at @e[type=armor_stand,tag=BOSSxiu,tag=!kuangfeng] run particle cloud ~ ~0.5 ~ 0.4 0.5 0.4 0.05 1 normal @a
execute at @e[type=armor_stand,tag=BOSSxiu,tag=!kuangfeng] run particle cloud ~ ~1.5 ~ 0.8 0.5 0.8 0.05 2 normal @a
execute at @e[type=armor_stand,tag=BOSSxiu,tag=!kuangfeng] run particle cloud ~ ~3 ~ 1.2 0.5 1.2 0.05 3 normal @a
execute at @e[type=armor_stand,tag=BOSSxiu,tag=!kuangfeng] run particle cloud ~ ~4.5 ~ 1.6 0.5 1.6 0.05 4 normal @a
execute at @e[type=armor_stand,tag=BOSSxiu,tag=!kuangfeng] run particle cloud ~ ~6 ~ 1.6 0.5 1.6 0.05 5 normal @a
execute at @e[type=armor_stand,tag=BOSSxiu,tag=!kuangfeng] run particle cloud ~ ~7.5 ~ 2 0.5 2 0.05 6 normal @a




#[大型狂风龙卷]其他实体凋零效果
execute at @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] positioned ~ ~2 ~ run effect give @e[tag=!BOSSxiu,distance=..6] wither 3 1 true
#[大型狂风龙卷]加速向休移动
execute as @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] at @s if entity @a[tag=BOSSxiu,distance=0..9] facing entity @a[tag=BOSSxiu,limit=1] feet run tp @s ^ ^ ^0.04
#[大型狂风龙卷]音效
# execute at @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] run playsound entity.ghast.shoot player @a ~ ~ ~ 2 2
execute at @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] run playsound entity.phantom.flap player @a ~ ~2 ~ 2 2
execute at @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] run playsound entity.phantom.flap player @a ~ ~2 ~ 2 1.5
execute at @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] run playsound entity.phantom.flap player @a ~ ~2 ~ 2 1
#[大型狂风龙卷]特效
execute at @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] run particle large_smoke ~ ~0.5 ~ 0.4 0.5 0.4 0.05 1 normal @a
execute at @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] run particle large_smoke ~ ~1.5 ~ 0.8 0.5 0.8 0.05 2 normal @a
execute at @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] run particle large_smoke ~ ~3 ~ 1.2 0.5 1.2 0.05 3 normal @a
execute at @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] run particle large_smoke ~ ~4.5 ~ 1.6 0.5 1.6 0.05 4 normal @a
execute at @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] run particle large_smoke ~ ~6 ~ 1.6 0.5 1.6 0.05 5 normal @a
execute at @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] run particle large_smoke ~ ~7.5 ~ 2 0.5 2 0.05 6 normal @a
#[大型狂风龙卷]破坏弹射物
execute at @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] positioned ~ ~2 ~ run kill @e[tag=yasuo,type=armor_stand,distance=..6]
execute at @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] positioned ~ ~2 ~ run kill @e[tag=!kuangfeng,type=#arrows,distance=..6]
execute at @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] positioned ~ ~2 ~ run kill @e[type=fireball,distance=..6]
execute at @e[type=armor_stand,tag=BOSSxiu,tag=kuangfeng] positioned ~ ~2 ~ run kill @e[type=snowball,distance=..6]
#[大型狂风龙卷]提升漂浮效果
execute at @e[type=minecraft:armor_stand,tag=BOSSxiu,tag=kuangfeng] run effect give @e[distance=0..2,tag=!BOSSxiu] levitation 1 4 true
execute at @e[type=minecraft:armor_stand,tag=BOSSxiu,tag=kuangfeng] run effect give @e[distance=2..4,tag=!BOSSxiu] levitation 1 3 true
execute at @e[type=minecraft:armor_stand,tag=BOSSxiu,tag=kuangfeng] run effect give @e[distance=4..6,tag=!BOSSxiu] levitation 1 2 true

#[狂风区域][龙卷风暴]到时间kill
kill @e[type=armor_stand,tag=kuangfengquyu,scores={skCDEntity=12..}]
kill @e[type=armor_stand,tag=BOSSxiu,scores={skCDEntity=10..}]

#C01休（风神祭祀）
execute as @a[tag=BossAlready,tag=BOSSxiu] run title @s actionbar [{"text":"狂风席卷","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"正在冷却：","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"score":{"objective":"skCDPearl","name":"@s"},"color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"s","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":" 狂风休止","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"正在冷却：","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"score":{"objective":"skCDEgg","name":"@s"},"color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"s","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":" 明岚千聚","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"正在冷却：","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"score":{"objective":"skCDSnowball","name":"@s"},"color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"s","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]
